

# FORTH-WINDOWS - Fenstertechnik für die Schneider CPCs

Claus Kühnel

Die Schneider CPCs sind bezogen auf ihre Rechnerklasse mit einer Vielzahl hervorragender Eigenschaften ausgestattet, die durch den BASIC-Befehlssatz gut unterstützt sind. Damit ist die Nutzung dieser Eigenschaften auf einfache Weise möglich.

**Mit der FORTH-Erweiterung WINDOWS sollte die einfache und wenig aufwendige Möglichkeit der Nutzung der komfortablen Eigenschaften der Schneider CPCs unter FORTH verdeutlicht werden.**

Mit dem cpcFORTH (Fa. FORTH-SYSTEME Angelika Flesch, Breisach) steht auch eine sehr komfortable FORTH-Implementierung für die Schneider CPCs zur Verfügung. Auch unter FORTH kann von den im Betriebssystem enthaltenen Routinen vorteilhaft Gebrauch gemacht werden. Anhand der FORTH-Erweiterung WINDOWS sollen Vorgehen und Anwendung betrachtet werden.

Im Dictionary des cpcFORTH sind die Fenstertechnik betreffend die Worte STREAM zur Umschal-

tung des aktiven Ein-/Ausgabegerätes und das Wort WINDOW zur Definition eines Textfensters im laufenden E/A-Gerät enthalten. Das Grafikfenster bezieht sich auf den ganzen Bildschirm. Der Ursprung läßt sich aber mit ORG beliebig setzen. Es liegt in der Natur von FORTH, daß man sich mit den gegebenen Eigenschaften nicht abfinden muß.

Eine Schlüsselstellung in der vorliegenden FORTH-Erweiterung nimmt das Definitionswort KOORD ein. Bei der Definition eines Textfensters werden die Fensternummer (Stream) und die Fensterkoordinaten einem Fensternamen zugeordnet. Beim Aufruf des so definierten Fensters übergibt das Wort die Daten auf den Stack. Das in Screen #7 enthaltene Demo verdeutlicht die Anwendung. Durch die Angabe 1 0 0 39 20 KOORD WINDOW1 wird ein Textfenster mit dem Namen WINDOW1 definiert, welches dem Stream #1 zugeordnet wird und sich über die Zeilen 0 bis 20 und die Spalten 0 bis 39 erstreckt. Die beim Aufruf auf dem Stack bereitgestellten Daten werden von den Worten FRAME und SELECT benötigt. Durch das Wort FRAME erfolgt ein Umranden des Textfensters, was dann zwangsläufig um diesen Rand verkleinert wird. Die um diesen Rand reduzierten Fensterkoordinaten hinterläßt das Wort wiederum auf dem Stack.

Erst SELECT definiert dann das eigentliche Textfenster. Es ergeben sich zwei Möglichkeiten der Definition. Soll das Fenster mit einem Rahmen versehen werden, dann muß der Aufruf beispielsweise durch WINDOW1 FRAME SELECT erfolgen. Nach dieser Wortfolge beziehen sich alle weiteren Textausgaben auf das ausgewählte Textfenster. Die Fensterumschaltung kann durch das Wort STREAM in Verbindung mit dem gewünschten Fensternamen & WINDOW1 STREAM) vorgenommen werden. Die überflüssigen Fensterkoordinaten und der von STREAM hinterlassene alte Wert werden vom Stack entfernt. Das ebenfalls in cpcFORTH enthaltene Wort CLS löscht das aktuelle Fenster und positioniert den Cursor links oben (Home Up). Soll auf das Löschen verzichtet werden, kann der Cursor durch HOMEUP positioniert werden.

Für die Definition eines Grafikfensters wurden vier neue PRIMITIVE WORDS definiert, die Bezug auf Routinen des Betriebssystems nehmen. Zum cpcFORTH mitgeliefert wird ein 8080/Z 80-Assembler, der die Definition von neuen PRIMITIVE WORDS ganz einfach über das Wort CODE zuläßt. Das Betriebssystem enthält u.a. Routinen zum Setzen der Breite und der Höhe des Grafikfensters. Fenstergrenzen werden automatisch auf Bildschirm-Bytengrenzen verlegt. Benutzt man später die Grenzen des Grafikfensters (z.B. bei der Normierung einer Wertefolge), sollten die aktuellen Grenzen, die durchaus von den vorgegebenen geringfügig abweichen können, abgefragt werden. Zur Formulierung der Grafikroutinen ist die Kenntnis der Adressen der Sprungtabelle und der Ein- bzw. Aussprungsbedingungen notwendig. Zum Setzen der Breite des Grafikfensters sind die linke und rechte Fensterbegrenzung in den Doppelregistern DE und HL zu übergeben. Bevor die Routine durch CALL BBCF aufgerufen wird, ist unbedingt das Doppelregister BC (Instruction Pointer) zu reiten. Die Routine zum Setzen der Höhe des Grafikfensters wird in der Sprungtabelle an der Adresse BBD2 vorgefunden. Der restliche Ablauf ist gleich. Das Verfahren bei der Abfrage der Fenstergrenzen ist praktisch umgekehrt. Die Routinen

# FORTH-WINDOWS

übergeben die Koordinaten in den Dopperegistern DE und HL, die dann nur noch auf den Stack gepusht werden müssen. Das Grafikfenster kann durch Aufruf des Wortes GRFRAME umrandet werden.

Der Abschluß bildet das in Screen #7 enthaltene Demo. Es werden zwei Textfenster und ein Grafikfenster definiert. Zur Ausgabe im


Textfenster dient das Wort NONSENS, welches 1000 über das Wort RANDOM zufällig bestimmte Zeichen an das ausgewählte Fenster sendet. Im Grafikfenster wird Rauschen dargestellt, welches ebenfalls mit dem Wort RANDOM erzeugt wird (siehe Bild unten).

Die vorgestellte Erweiterung belegt 563 Byte (ohne Demo) und ist damit sehr kompakt. Durch die

Verbindung von Text- und Grafikfenstern ergeben sich günstige Möglichkeiten für die Gestaltung eines Dialoges mit dem Bediener.

Dr. Claus Kühnel  
Zschertnitzer Str. 52  
DDR-8020 Dresden

Registriernummer des Büros für  
Urheberrechte der DDR: VZA-  
2052-88



Screen # 0  
WINDOWS:

CK 20/ 7/88)

Die FORTH-Erweiterung WINDOWS erlaubt die von BASIC her gewohnte Fenstertechnik mit den Schneider CPCs.

Verwendet wurde das cpcFORTH der  
Fa. FORTH-SYSTEME Angelika Flesch PF 1103 7814 Breisach.

Zur Verdeutlichung der Anwendung ist in Screen #7 ein Demo  
enthalten.

# FORTH-WINDOWS

Screen # 1  
( WINDOWS: Load Screen

CK 20/ 7/88)

DECIMAL

CR SP@ ^STR - U. .( bytes in dictionary )

CR CR .( Wait ... loading WINDOWS ) CR CR

2 7 THRU

.( WINDOWS loaded. )

CR CR SP@ ^STR - U. .( bytes left in dictionary )

CR CR

Screen # 2  
( WINDOWS: KOORD

CK 20/ 7/88)

```
: KOORD CREATE C, C, C, C, C,  
  DOES> DUP C@ SWAP 1+  
         DUP C@ SWAP 1+  
         DUP C@ SWAP 1+  
         DUP C@ SWAP 1+ C@ ;
```

( Definition von Textfenstern:

Fenster# xlinks yoben xrechts yunten KOORD Fenstername )

Z-80 cpcForth 1.1 (Forth-83) by Forth-Systeme A. Flesch 1985

# FORTH-WINDOWS

Screen # 3  
( WINDOWS: FRAME

CK 20/ 7/88)

```
: FRAME      ( yu xr yo xl Fenster# -- yu' xr' yo' xl' Fenster# )
  0 STREAM DROP
  OVER 3 PICK GOTOXY 150 EMIT
  3 PICK 2 PICK 1+ DO 154 EMIT LOOP 156 EMIT
  4 PICK 3 PICK 1+ DO
    OVER 1 DUP ROT SWAP GOTOXY 149 EMIT
    4 PICK SWAP          GOTOXY 149 EMIT LOOP
  OVER 5 PICK GOTOXY 147 EMIT
  3 PICK 2 PICK 1+ DO 154 EMIT LOOP 153 EMIT
  4 ROLL 1- 4 ROLL 1- 4 ROLL 1+ 4 ROLL 1+ 4 ROLL ;
```

Screen # 4  
( WINDOWS: >STREAM SELECT HOMEUP

CK 20/ 7/88)

```
: >STREAM      ( yu xr yo xl Fenster# -- )
  STREAM DROP 2DROP 2DROP ;

: SELECT      ( yu xr yo xl Fenster# -- )
  STREAM DROP
  ROT SWAP 2SWAP WINDOW ;

: HOMEUP      ( -- )
  0. GOTOXY ;
```

Screen # 5  
( WINDOWS: GRAFIK1

CK 20/ 7/88)

HEX

```
BB0F CONSTANT ADDRSETWIDTH
BB02 CONSTANT ADDRSETHEIGHT ( Firmwareadressen )
BB05 CONSTANT ADDRGETWIDTH ( Schneider CPC 6128 )
BB08 CONSTANT ADDRGETHEIGHT
```

DECIMAL

```
CODE SETWIDTH
  D POP  H POP  B PUSH ADDRSETWIDTH CALL  B POP  NEXT
```

```
CODE SETHEIGHT
  D POP  H POP  B PUSH ADDRSETHEIGHT CALL  B POP  NEXT
```

# FORTH-WINDOWS

Screen # 6

( WINDOWS: GRAFIK2

CK 20/ 7/88)

CODE GETWIDTH

B PUSH ADDRGETWIDTH CALL B POP D PUSH H PUSH NEXT

CODE GETHEIGHT

B PUSH ADDRGETHEIGHT CALL B POP H PUSH D PUSH NEXT

```
: GRFRAME ( -- )
  GETWIDTH GETHEIGHT 3 PICK 2 PICK MOVE
  2 PICK 2 PICK DRAW 2 PICK OVER DRAW
  3 PICK OVER DRAW 3 PICK 2 PICK DRAW
  2DROP 2DROP ;
```

Screen # 7

( WINDOWS: DEMO

CK 20/ 7/88)

1 0 0 39 20 KOORD WINDOW1

( Definition von zwei )

2 0 21 79 24 KOORD WINDOW2

( Textfenstern )

320 630 SETWIDTH

( Setzen der Koordinaten )

70 390 SETHEIGHT

( des Grafifensters )

: NOISE 641 320 DO 1 350 RANDOM 50 + DRAW 2 +LOOP ;

: NONSENS 1000 0 DO 256 RANDOM EMIT LOOP ;

: DEMO

WINDOW1 FRAME SELECT CLS NONSENS CR ." WINDOW 1"

WINDOW2 FRAME SELECT CLS NONSENS CR ." WINDOW 2"

GRCLS GRFRAME NOISE 1 STREAM DROP ;

: RUNDEMO 10 0 DO DEMO LOOP ;

Screen # 8